Задача 2. Определение формы препятствия

Задача. На пути у одинокого путника, идущего в глубоком тумане по экрану монитора встречается препятствие, стенки которого параллельны краям монитора, требуется определить форму препятствия. Путешественник не ориентируется в пространстве, у него нет карты, и он не может запоминать точки, в которых он был, но он знает, где у него правая рука, а где левая. 

Особенности задачи. 

1. Из условия ясно, что путник не может использовать свои координаты. Таким образом, для принятия решения о том, обошёл ли он всё препятствие, путник может использовать только типы (каждой стенки по определённому правилу присваивается число, которое и называется её типом, правило определения типа видно из алгоритма) пройденных стенок. Всего типов стенок 4. 

2. Под формой препятствия подразумевается последовательность типов стенок препятствия.

Главная идея решения. Обходя препятствие, путник записывает тип текущей стенки в  массив. Обойдя препятствие ровно два раза (путник делает вывод о том, что обход выполнен дважды, исходя из того, что первая половина массива типов стенок полностью идентична его второй половине) и остановившись на стенке № 4 (в любом случае последняя стенка препятствия имеет тип 4, это следует из логики алгоритма), путник может сказать, что препятствие было точно им обойдено. Стенка, с которой путник начинает движение, всегда имеет тип 1.

Построение теста. Первоначально на экране строится геометрическая фигура (препятствие) из линий, параллельных стенкам экрана. Затем выбирается некоторая точка на препятствии, с которой начинается движение.

Виды поворотов. Внешний поворот – 


Внутренний поворот – 

Алгоритм. 

Пока путник не выяснит форму препятствия, повторять
1. Если путник обнаруживает, что перед ним нет стены, то он совершает внешний поворот, после которого тип новой стенки определяется как тип предыдущей стенки + 1 (но 4+1=1).

2. Если на пути путника возникает стена, то он совершает внутренний поворот, после которого тип новой стенки определяется как тип предыдущей стенки – 1  (но 1–1=4).

3. Иначе продолжать движение в прежнем направлении.

Текст программы.

program forma;

uses crt, graph;

var T: array [1..1000] of 0..4; k: integer;

    i, j, x, y, N: integer;

procedure init;

var gd, gm: integer;

begin

 gd:=detect; initgraph(gd, gm, '');

 moveto(50,50);

 lineto(60, 50); lineto(60,60); lineto(70,60); lineto(70,70);

 lineto(80,70); lineto(80,80); lineto(50,80); lineto(50,50);

 for i:=getmaxX downto 1 do

  for j:=getmaxY downto 1 do

   if getpixel(i,j)<>0 then

    begin

     x:=i; y:=j;

    end;

 k:=1; T[k]:=1;

 outtextXY(10,10,'Форма препятствия:');

end;

function konets: boolean;

var i: integer; f: boolean; S: string;

begin

 f:=false;

 if (T[k-1]=4) and(k mod 2 <> 0) then

  begin

   f:=true;

   for i:=1 to (k-1) div 2 do

    if T[i]<>T[i+(k-1) div 2] then f:=false;

  end;

 konets:=f;

 str(T[k],S);

 if not f then outtextXY(10,10+10*k, S);

end;

procedure povorot1;

begin

 if T[k]=4 then T[k+1]:=1 else T[k+1]:=T[k]+1; k:=k+1;

end;

procedure povorot2;

begin
 if T[k]=1 then T[k+1]:=4 else T[k+1]:=T[k]-1; k:=k+1;

 case T[k] of

  1: begin putpixel(x+1,y,random(14)+1); x:=x+1; end;

  2: begin putpixel(x,y+1,random(14)+1); y:=y+1; end;

  3: begin putpixel(x-1,y,random(14)+1); x:=x-1; end;

  4: begin putpixel(x,y-1,random(14)+1); y:=y-1; end;

 end;

end;

procedure go;

begin
 case T[k] of

  1: begin x:=x+1; putpixel(x,y,random(14)+1); end;

  2: begin y:=y+1; putpixel(x,y,random(14)+1); end;

  3: begin x:=x-1; putpixel(x,y,random(14)+1); end;

  4: begin y:=y-1; putpixel(x,y,random(14)+1); end;

 end;

end;

begin

 init;

 while not konets do begin

  case T[k] of

   1: if getpixel(x,y-1)<>0 then povorot2

      else if getpixel(x+1,y)=0 then povorot1

      else go;

   2: if getpixel(x+1,y)<>0 then povorot2

      else if getpixel(x,y+1)=0 then povorot1

      else go;

   3: if getpixel(x,y+1)<>0 then povorot2

      else if getpixel(x-1,y)=0 then povorot1

      else go;

   4: if getpixel(x-1,y)<>0  then povorot2

      else if getpixel(x,y-1)=0 then povorot1

      else go;

  end; delay(500); if keypressed then halt(1); end;

 outtextXY(10,20+10*k,'СТОП’);

 repeat until keypressed;

 closegraph;

 writeln('Форма препятствия:');

 for i:=1 to (k-1) div 2 do

  write(T[i]);

 readkey; repeat until keypressed;

end.

